12. Representation of polynomials and rational expressions

12.1 Internal representation of polynomials

There are three levels of selecting data from a polynomial. At the highest level we have **coeffs** and **coeff**. The commands **nops** and **op** work at a lower level. Finally, at the lowest level, we use **dismantle** to get to the Directed Acyclic Graph representation of a polynomial.

```
[> p := x^10 + 2*x^4 + 9;
[> coeffs(p);              # sequence of coefficients
[> whattype(p);
```

While p is a polynomial, it is also a general expression of type + on which the usual selectors work:

```
[> nops(p);                # number of terms
[> op(p);                  # sequence of terms
```

The op command allows to select deeper into the polynomial:

```
[> t2 := op(2,p);          # select second term of p
```

A term consists of a coefficient and a monomial:

```
[> op(t2);
[> mon := op(2,t2);
```

A monomial consists of a name for a variable and an exponent:

```
[> op(mon);
[> op(2,mon);
```

Instead of the assignments to t2 and mon, we can select the exponent of the monomial in the second term of p all at once:

```
[> op(2,op(2,op(2,p)));
```

Alternatively, a list of indices can be provided as first argument to the op command:

```
[> op([2,1],p); op([2,2,1],p); op([2,2,2],p);
```

A detailed view of polynomials can be obtained with dismantle (see Figure 1):

```
[> dismantle(p);
```



Figure 1: Directed Acyclic Graph of $x^{10} + 2x^4 + 9$.

Notice: in storing the terms in the sum, the monomials come first, followed by the coefficients. So the polynomial $x^{10} + 2x^4 + 9$ is actually stored like $x^{10} \times 1 + x^4 \times 2 + 9 \times 1$.

Now we understand the structure, and knowing that Maple stores every object only once, we can explain the effects of a substitution like this (see Figure 2):

```
[> q := subs(1=Pi,p);
[> dismantle(q);
```



Figure 2: Directed Acyclic Graph of $x^{10}\pi + 2x^4 + 9\pi$.

Observe that `protected` is an attribute to the name `Pi`. When a variable has no attribute (like `x`), then this field of the name simply points to `NULL`.

Since the polynomial $\texttt{p} = x^{10} + 2x^4 + 9$ is stored like $x^{10} \times 1 + x^4 \times 2 + 9 \times 1$ and every object is stored only once, we now see why the command `subs(1=Pi,p)` creates the polynomial $x^{10} \times \pi + x^4 \times 2 + 9 \times \pi$.

12.2 Generalized rational expressions

Maple can work with polynomials where the variable is replaced by, for example a cosine:

```
[> pcos := subs(x=cos(x),p);
[> whattype(pcos); type(pcos,polynom);
```

While the expression is no longer a true polynomial, we can still ask for the coefficients :

```
[> coeffs(pcos);
```

But we can no longer factor the polynomial directly :

```
[> factor(pcos,complex);
```

We have to work indirectly, first factor p and then substitute x by fp in the factored polynomial:

```
[> fp := factor(p,complex);
[> fpcos := subs(x=cos(x),fp);
```

General rational expressions are rational polynomial in disguise:

```
[> rp := ((sin(y))^2-1)/(sin(y)-1);
[> type(rp,ratpoly);
[> factor(numer(rp));
[> normal(rp);
```

It is interesting to see the internal structure (see Figure):

```
[> dismantle(rp);
```



Figure 3: Directed Acyclic Graph of $\frac{\sin(y)^2-1}{\sin(y)-1}$.

12.3 Assignments

1. For the polynomial $p = x^3 - 8x + 1$, give the Maple command (only one command!) to select the coefficient -8.

2. Use dismantle to investigate the internal representation of the polynomial $2x(y^2+1)^2$. From the output of dismantle, draw the directed acyclic graph of this polynomial.

3. Consider the following instructions

```
[> x^2 - x + 1/x - 1/x^2;
[> subs(-1=1,%);
```

   Explain the result of this substitution.